
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 15.1.1 which is part of the SSNITZ test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_15_1_1
Test Group:
SS NITZ
ATS Version:
iwd-TTCN3-B2013-03_D14wk24 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Samsung Galaxy S3 UE
Verification Status:
PASS
4 Corrections required for test case 15.1.1
4.1 Introduction

This section describes the changes required to make test case 15.1.1 run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D14wk24 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 Module MTC_Component_SSNITZ
	Record Name
	MTC_Component_SSNITZ

	Reason for change
	During the clean up SS NITZ test cases the execution does not terminate the test case correctly, as a results the test case only terminates due to guard timer expiry with a fail verdict

	Summary of change
	Added MTC_Components_Type and  introduced vc_Components to handle UTRA component

	Source of change
	MTC_Component_SSNITZ.ttcn


Before change: 

	module MTC_Component_SSNITZ {
  import from MTC_Base all;
  import from UTRAN_Component all;
  /*--------------------------------------------------------------------------*/
  // local type definitions:
  // Component:
  type component MTC_SSNITZ extends MTC_BASE {};
  type component SYSTEM_SSNITZ extends MTC_BASE_SYSTEM, UTRAN_SYSTEM {};
}



After change: 

	module MTC_Component_SSNITZ {
  import from MTC_Base all;
  import from UTRAN_Component all;
  /*--------------------------------------------------------------------------*/
  // local type definitions:
  //WA#15_1_1 Added MTC component type UTRA
  type record MTC_Components_Type {
  UTRAN_PTC  Utran
  };
  // Component:
  type component MTC_SSNITZ extends MTC_BASE {
   var MTC_Components_Type vc_Components;
 //WA#15_1_1 Added to handle utra component
  };
  type component SYSTEM_SSNITZ extends MTC_BASE_SYSTEM, UTRAN_SYSTEM {};
}



4.3 Module MTC_Main_SSNITZ
	Record Name
	MTC_Main_SSNITZ

	Reason for change
	1) At the start of the test case Switch on the UE command is not sent correctly as the UT port mapping between PTC and MTC and not connected

2) During the clean up SS NITZ test cases the execution does not terminate the test case correctly, as a results the test case only terminates due to guard timer expiry 

	Summary of change
	1)  In f_MTC_ConnectPTCs_SSNITZ added connect(mtc:PTC_Ut[tsc_MTC_PortIndex_UTRAN], p_Utran:UT);
2) Modified all function to be run on MTC_SSNITZ, as vc_Components was introduced.
3)  In f_MTC_ConnectPTCs_SSNITZ vc_Components.Utran is updated
4) fl_MTC_StopIfFinished is updated to check if vc_Components.Utran == null
5) a_PTC_Control is updated to handle [vc_Components.Utran != null]

	Source of change
	MTC_Main_SSNITZ.ttcn


Before change: 

	module MTC_Main_SSNITZ {
  import from MTC_Base all;
  import from MTC_UpperTester all;
  import from UpperTesterSystemCmds all;
  import from UTRAN_Component all;
  import from CommonDefs all;
  //============================================================================
  /*
   * @desc      Connect all PTCs
   * @param     p_Utran
   * @status
   */
  function f_MTC_ConnectPTCs_SSNITZ(UTRAN_PTC p_Utran) runs on MTC_BASE
  {
    if (p_Utran == null) {
      FatalError(__FILE__, __LINE__, "invalid configuration");
      mtc.stop;
    }
    // ***** MTC *****
    f_MTC_BASE_Map();
    // ***** UTRAN PTC *****
    f_UTRAN_PTC_Map(p_Utran);
    f_UT_InitialPowerOffUE(Ut);
  }
  //----------------------------------------------------------------------------
  /*
   * @desc      stop mtc, NAS emulator, IP PTC etc. when all test case components have finished
   */
  function fl_MTC_StopIfFinished()
  {
    f_MTC_Stop(pass);                      // 'pass' will overrule 'none' only
  }
  //----------------------------------------------------------------------------
  /*
   * @desc      Interface to control the other components
   * @status
   */
  altstep a_PTC_Control()
  { /* NOTEs:
       - acc. to ES 201 873-1 Annex F.1.1 an error verdict shall be assignd when "running" is applied on any uninitialised component
       - it seems that "running" cannot be used in the "[]" */
    [] any component.killed
      { // at a component something's gone wrong and it killed itself, so have to stop everything else
        all component.kill;
        mtc.stop;
      }
  }
  //============================================================================
  /*
   * @desc      Main loop of the MTC
   * @param     p_GuardTimer timer
   */
  function f_MTC_MainLoop(timer p_GuardTimer) runs on MTC_BASE
  { /* @sic R5s120520: f_UT_InitialPowerOffUE moved to f_MTC_ConnectPTCs_POS sic@ */
    f_MTC_BASE_Init();
    while (true) {      /* mtc will be stopped by a_PTC_Control or due to timeout of the guard timer;
                           => neither a_PTC_Control nor a_UpperTester needs to use 'repeat' */
      alt {
        [] a_PTC_Control();             // handle 'done' and 'killed' of PTCs: MTC will be stopped, when all PTCs are terminated or a fatal error occurs
        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_UTRAN);    // handle upper tester for PTC 1 (UTRAN)
        [] a_MTC_ReceiveFromUT_SendToPTC();
        [] p_GuardTimer.timeout {f_MTC_Stop(fail);}    /* acc. to the assumption that preamble and postamble are short compared to the testbody
                                                          a FAIL verdict is assigned (rather than an INCONC) */
      }
    }
  }
}



After change: 

	module MTC_Main_SSNITZ {
  import from MTC_Base all;
  import from MTC_Component_SSNITZ all;
  import from MTC_UpperTester all;
  import from UpperTesterSystemCmds all;
  import from UTRAN_Component all;
  import from CommonDefs all;
  //============================================================================
  /*
   * @desc      Connect all PTCs
   * @param     p_Utran
   * @status
   */
  function f_MTC_ConnectPTCs_SSNITZ(UTRAN_PTC p_Utran) runs on MTC_SSNITZ //WA#15_1_1 changed to SSNITZ componen
  {
   vc_Components.Utran := p_Utran; //WA#15_1_1
    if (p_Utran == null) {
      FatalError(__FILE__, __LINE__, "invalid configuration");
      mtc.stop;
    }
    // ***** MTC *****
    f_MTC_BASE_Map();
    // ***** UTRAN PTC *****
    f_UTRAN_PTC_Map(p_Utran);
    // ***** Connect PTCs to MTC *****
    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_UTRAN], p_Utran:UT); //WA#15_1_1 map Ut port
    f_UT_InitialPowerOffUE(Ut);
  }
  //----------------------------------------------------------------------------
  /*
   * @desc      stop mtc, NAS emulator, IP PTC etc. when all test case components have finished
   */
  function fl_MTC_StopIfFinished() runs on MTC_SSNITZ //WA#15_1_1 changed to SSNITZ component
  {
     if (vc_Components.Utran == null){ //WA#15_1_1
    f_MTC_Stop(pass);                      // 'pass' will overrule 'none' only
  }
  }
  //----------------------------------------------------------------------------
  /*
   * @desc      Interface to control the other components
   * @status
   */
  altstep a_PTC_Control() runs on MTC_SSNITZ //WA#15_1_1 changed to SSNITZ component
  { /* NOTEs:
       - acc. to ES 201 873-1 Annex F.1.1 an error verdict shall be assignd when "running" is applied on any uninitialised component
       - it seems that "running" cannot be used in the "[]" */
    [] any component.killed
      { // at a component something's gone wrong and it killed itself, so have to stop everything else
        all component.kill;
        mtc.stop;
      }
    [vc_Components.Utran != null] vc_Components.Utran.done //WA#15_1_1 update vc_Component.
      {
        vc_Components.Utran := null;
        fl_MTC_StopIfFinished();
      }
  }
  //============================================================================
  /*
   * @desc      Main loop of the MTC
   * @param     p_GuardTimer timer
   */
  function f_MTC_MainLoop(timer p_GuardTimer) runs on MTC_SSNITZ //WA#15_1_1 changed to SSNITZ component
  { /* @sic R5s120520: f_UT_InitialPowerOffUE moved to f_MTC_ConnectPTCs_POS sic@ */
    f_MTC_BASE_Init();
    while (true) {      /* mtc will be stopped by a_PTC_Control or due to timeout of the guard timer;
                           => neither a_PTC_Control nor a_UpperTester needs to use 'repeat' */
      alt {
        [] a_PTC_Control();             // handle 'done' and 'killed' of PTCs: MTC will be stopped, when all PTCs are terminated or a fatal error occurs
        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_UTRAN);    // handle upper tester for PTC 1 (UTRAN)
        [] a_MTC_ReceiveFromUT_SendToPTC();
        [] p_GuardTimer.timeout {f_MTC_Stop(fail);}    /* acc. to the assumption that preamble and postamble are short compared to the testbody
                                                          a FAIL verdict is assigned (rather than an INCONC) */
      }
    }
  }
}


4.4 f_TC_15_1_1_UTRAN
	Record Name
	f_TC_15_1_1_UTRAN()

	Reason for change
	In the current TTCN implementation of this function, the function f_UTRAN_CC_Disconnect is called. Unless a specific TI value is passed as an argument to this, it always uses cs_TI_MO as the TI. 

In the current situation, there is a MT call being handled, and hence the TI value used should be cs_TI_MT.
Also, in the current TTCN implementation the function f_UTRAN_SwitchPower_Off is used for switch off and detach. This function handles a situation when a service REQ is received etc. A better way to would be to use the function f_UTRAN34_SwitchPower_Off instead.

	Summary of change
	In the call to function f_UTRAN_CC_Disconnect , passed cs_TI_MT as an argument.

Also, instead of f_UTRAN_SwitchPower_Off used f_UTRAN34_SwitchPower_Off.

(Please see screenshots)

	Source of change
	CLIP_CLIR_CNAP.ttcn


Before:

	// @siclog "Step 10" siclog@
    if (f_UT_ChkCallDisplayed(UT, v_CallingNumber)) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");
    }
    f_UTRAN_TestBody_Set(false);
    f_UTRAN_CC_Disconnect (utran34_Cell1);
    f_UTRAN_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);
    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));


After:

	// @siclog "Step 10" siclog@
    if (f_UT_ChkCallDisplayed(UT, v_CallingNumber)) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");
    }
    f_UTRAN_TestBody_Set(false);
    //WA#15_1_1 f_UTRAN_CC_Disconnect(utran34_Cell1);
    f_UTRAN_CC_Disconnect (utran34_Cell1,cs_TI_MT);//WA#15_1_1 Added cs_TI_MT
    //WA#15_1_1 f_UTRAN_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);
    f_UTRAN34_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);
    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));


4.5 f_UTRAN34_SwitchPower_Off
	Record Name
	f_UTRAN34_SwitchPower_Off

	Reason for change
	In the current TTCN implementation of this function, two cases are handled : the UE is in idle mode  or UE is in connected mode.  For this testcase the execution goes in the connected mode path.

In this path, the DETACH REQ is expected only as a UplinkDirectTransfer message. This needs to be extended to handle InitdirectTransfer as well.

	Summary of change
	Implemented the handling for InitDirectTransfer in the connected mode path. Please see screenshot for details.

	Source of change
	UTRAN34_CommonFunctions.ttcn


Before:

	else { // UE is in connected mode; @sic R5s110155 sic@
        alt {
          //Step 5a1, 5c1 & 5d1: Receive Detach Request message for the different scenarios
          // - NMO1 and (C1 or C3) : detach type = power switched off, GPRS/IMSI combined detach
          // - (NMO1 or NMO2) and C2 : detach type = power switched off, PS detach
          // - NMO2 and (C1 or C3) : detach type = power switched off, PS detach
          [] U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                                      tsc_RB3,
                                                      cr_U_DetachRequest))                          
            {
              //Step 5d2: Receive IMSI Detach indication
              // - NMO2 and (C1 or C3)
              if (v_AttFlag == 1 and
                  ((v_Nmo == tsc_NMO_II and (v_Utra_Postamble_Condition == UTRA_C1 or v_Utra_Postamble_Condition == UTRA_C3)))) { // @sic R5s110155, R5-115770 sic@
                U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,  // @sic R5s110155, R5s110695, R5s120529 sic@
                                                    tsc_RB3,
                                                    cr_U_Imsi_DetachInd));
              }
            }


After:
	else { // UE is in connected mode; @sic R5s110155 sic@
        alt {
          //Step 5a1, 5c1 & 5d1: Receive Detach Request message for the different scenarios
          // - NMO1 and (C1 or C3) : detach type = power switched off, GPRS/IMSI combined detach
          // - (NMO1 or NMO2) and C2 : detach type = power switched off, PS detach
          // - NMO2 and (C1 or C3) : detach type = power switched off, PS detach
          [] U_Dc.receive((car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                                      tsc_RB3,
                                                      cr_U_DetachRequest), // @sic R5s110594 sic@
                          //WA#15_1_1 : Start change : Added the car_PS_InitDirectTransfer handling
                          car_PS_InitDirectTransfer(utran_CellDedicated,
                                                      tsc_RB3,
                                                      cr_U_DetachRequest)))                          
            {
              //Step 5d2: Receive IMSI Detach indication
              // - NMO2 and (C1 or C3)
              if (v_AttFlag == 1 and
                  ((v_Nmo == tsc_NMO_II and (v_Utra_Postamble_Condition == UTRA_C1 or v_Utra_Postamble_Condition == UTRA_C3)))) { // @sic R5s110155, R5-115770 sic@
                U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,  // @sic R5s110155, R5s110695, R5s120529 sic@
                                                    tsc_RB3,
                                                    cr_U_Imsi_DetachInd));
              }
            }


4.6 f_UTRAN_CC_Disconnect
	Record Name
	f_UTRAN_CC_Disconnect

	Reason for change
	In the current TTCN implementation of f_UTRAN_CC_Disconnect, the cell state cell_DCH_StandAloneSRB is not handled. This needs to be implemented.

	Summary of change
	Handled the cell state cell_DCH_StandAloneSRB in the function f_UTRAN_CC_Disconnect. Please see screenshot.

	Source of change
	UTRAN_CommonFunctions.ttcn


Before:

	function f_UTRAN_CC_Disconnect(UTRAN_CellId_Type p_CellId,
                                 template (value) TI  p_TI_S:= cs_TI_MO) runs on UTRAN_PTC
  {
    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    if (v_CellConfig == cell_DCH_Speech or v_CellConfig == cell_Four_DTCH_PS_CS) { // @sic R5s120070 sic@
      // Step 1: Transmit Downlink Direct Transfer with a Disconnect message
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_Disconnect(p_TI_S))); // @sic R5s110547 sic@
      // Step 2: Receive Uplink Direct Transfer with a Release message
      U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_Rel(cr_TI_Any)));
      // Step 3: Transmit Downlink Direct Transfer with a Release Complete message
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3,  cs_U_RelCmpl(p_TI_S))); // @sic R5s110547 sic@
      //FFS - SS configuration and update of Cellconfig.
    }
  }


After:
	function f_UTRAN_CC_Disconnect(UTRAN_CellId_Type p_CellId,
                                 template (value) TI  p_TI_S:= cs_TI_MO) runs on UTRAN_PTC
  {
    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    if (v_CellConfig == cell_DCH_Speech or v_CellConfig == cell_Four_DTCH_PS_CS or v_CellConfig == cell_DCH_StandAloneSRB) { // @sic R5s120070 sic@
      //WA#15_1_1 : Added v_CellConfig == cell_DCH_StandAloneSRB above in the if statement
      // Step 1: Transmit Downlink Direct Transfer with a Disconnect message
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_Disconnect(p_TI_S))); // @sic R5s110547 sic@
      // Step 2: Receive Uplink Direct Transfer with a Release message
      U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_Rel(cr_TI_Any)));
      // Step 3: Transmit Downlink Direct Transfer with a Release Complete message
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3,  cs_U_RelCmpl(p_TI_S))); // @sic R5s110547 sic@
      //FFS - SS configuration and update of Cellconfig.
    }
  }


5 Branches executed in test case 15.1.1
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Samsung Galaxy S3 UE
The Samsung Galaxy S3 UE passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Log masking as per RAN5 PRD12 was applied to prevent any disclosure of confidential UE information.

Test Case Execution log file: 
Rohde-Schwarz\tc_15_1_1_Samsung_S3.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
7 References

	R5s140612
This archive comprises HTML Execution log files, PICS/PIXIT files and the TTCN MP file


